# AFFINE ISOMETRIES IN THE SPACE.

## SYMMETRIES

Let $E_3 = R^3$ be the euclidean affine plane, with the standard scalar product.

Unless otherwise stated we will work with the standard coordinate system $\mathrm{R} = \{\mathrm{O}, \mathrm{B} = \{e_1, e_2, e_3\}\}$, which is an orthonormal system. We denote by $(x, y, z)$ the coordinates of an arbitrary poin in $E_3$.

We will construct the matrix $M_f(\mathrm{R})$ of a symmetry $\mathrm{f}: E_3 \to E_3$ from the matrices of the next basic symmetries :

| Plane of symmetry (plane of fixed points) | x=0 | y=0 | z=0 |
|---|---|---|---|
| $M_f(\mathrm{R})$ | $\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$ | $\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$ | $\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix}$ |

```
> with(plots):
> plane:=implicitplot3d(x=0,x=-1..1,y=-1..1,z=-1..1,color=yellow,
  style=surface):
> ge1:=arrow(<1,0,0>, shape = arrow,color=blue):gfe1:=arrow(<-1,0,
  0>, shape = arrow,color=red):
> ge2:=arrow(<0,1,0>, shape = arrow,color=blue):gfe2:=arrow(<0,1,
  0>, shape = arrow,color=red):
> ge3:=arrow(<0,0,1>, shape = arrow,color=blue):gfe3:=arrow(<0,0,
  1>, shape = arrow,color=red):
> display(plane,ge1,ge2,ge3,gfe1,gfe2,gfe3);
```



## Determine the equations of the reflection with respect to the plane $x+2z-2=0$ in the orthonormal coordinate system R={O,B={$e_1$, $e_2$, $e_3$}}.

```
> restart; with(linalg):
> solve({x+2*z-2 = 0},{x,y,z});
```

$$\{x=-2z+2, y=y, z=z\} \quad (1.1)$$

```
> Q:=[2,0,0]; v1:=[0,1,0]; v2:=[-2,0,1];
```

$$Q := [2, 0, 0]$$

$$v1 := [0, 1, 0]$$

$$v2 := [-2, 0, 1] \quad (1.2)$$

Normal vector to the plane.

```
> n:=[1,0,2];
```

$$n := [1, 0, 2] \quad (1.3)$$

```
> innerprod(v1,v2);
```

$$0 \tag{1.4}$$

We need an orthonormal coordinate system R'={Q,{u1,u2,u3}}. The previous 3 vector are already pairwise orthogonal, so we make them unitary.

```
> u[1]:= evalm(n/(sqrt(innerprod(n,n))));u[2]:=v1/(sqrt(innerprod
  (v1,v1))); u[3]:=v2/(sqrt(innerprod(v2,v2)));
```

$$u_1 := \left[\begin{array}{ccc} \frac{1}{5}\sqrt{5} & 0 & \frac{2}{5}\sqrt{5} \end{array}\right]$$

$$u_2 := [0, 1, 0]$$

$$u_3 := \frac{1}{5}\,[-2, 0, 1]\sqrt{5} \tag{1.5}$$

The matrix of the symmetry with respect to the plane z1=0 in the coordinate system R' is:

```
> Sp:=matrix([[1,0,0,0],[0,-1,0,0],[0,0,1,0],[0,0,0,1]]);
```

$$Sp := \left[\begin{array}{cccc} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{array}\right] \tag{1.6}$$

Matrices of change of coordinates:

```
> Qm:=matrix(3,1,Q);U1:=matrix(3,1,u[1]);U2:=matrix(3,1,evalm(u
  [2]));U3:=matrix(3,1,evalm(u[3]));
```

$$Qm := \left[\begin{array}{c} 2 \\ 0 \\ 0 \end{array}\right]$$

$$U1 := \left[\begin{array}{c} \frac{1}{5}\sqrt{5} \\ 0 \\ \frac{2}{5}\sqrt{5} \end{array}\right]$$

$$U2 := \left[\begin{array}{c} 0 \\ 1 \\ 0 \end{array}\right]$$

$$\tag{1.7}$$

$$U3 := \begin{bmatrix} -\frac{2}{5}\sqrt{5} \\ 0 \\ \frac{1}{5}\sqrt{5} \end{bmatrix} \quad (1.7)$$

```
> RpR:=stackmatrix(matrix([[1,0,0,0]]),concat(Qm,U1,U2,U3));
```

$$RpR := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 2 & \frac{1}{5}\sqrt{5} & 0 & -\frac{2}{5}\sqrt{5} \\ 0 & 0 & 1 & 0 \\ 0 & \frac{2}{5}\sqrt{5} & 0 & \frac{1}{5}\sqrt{5} \end{bmatrix} \quad (1.8)$$

The matrix of the symmetry in the coordinate system R is:

```
> S:=evalm(RpR&*Sp&*inverse(RpR));
```

$$S := \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{4}{5} & \frac{3}{5} & 0 & -\frac{4}{5} \\ 0 & 0 & 1 & 0 \\ \frac{8}{5} & -\frac{4}{5} & 0 & -\frac{3}{5} \end{bmatrix} \quad (1.9)$$

```
> evalm(S&*matrix(4,1,[1,x,y,z]))=matrix(4,1,[1,x1,y1,z1]);
```

$$\begin{bmatrix} 1 \\ \frac{4}{5}+\frac{3}{5}x-\frac{4}{5}z \\ y \\ \frac{8}{5}-\frac{4}{5}x-\frac{3}{5}z \end{bmatrix} = \begin{bmatrix} 1 \\ x1 \\ y1 \\ z1 \end{bmatrix} \quad (1.10)$$

## Classify the isometry with matrix M in the coordinate system R={O,B={$e_1$, $e_2$, $e_3$}}.

## b) Obtain its invariant subspaces and use them to decompose it.

```
> restart; with(linalg):
> M:=matrix([[1, 0, 0, 0], [-1/5, 3/5, 0, -4/5], [1, 0, 1, 0],
  [8/5, -4/5, 0, -3/5]]);
```

$$M := \begin{bmatrix} 1 & 0 & 0 & 0 \\ -\frac{1}{5} & \frac{3}{5} & 0 & -\frac{4}{5} \\ 1 & 0 & 1 & 0 \\ \frac{8}{5} & -\frac{4}{5} & 0 & -\frac{3}{5} \end{bmatrix} \quad (2.1)$$

```
> A:=submatrix(M,2..4,2..4); b:=submatrix(M,2..4,1..1);
```

$$A := \begin{bmatrix} \frac{3}{5} & 0 & -\frac{4}{5} \\ 0 & 1 & 0 \\ -\frac{4}{5} & 0 & -\frac{3}{5} \end{bmatrix}$$

$$b := \begin{bmatrix} -\frac{1}{5} \\ 1 \\ \frac{8}{5} \end{bmatrix} \quad (2.2)$$

```
> evalm(A&*transpose(A));det(A);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$-1 \quad (2.3)$$

```
> AI:=evalm(A-diag(1,1,1)); rank(AI); rank(concat(b,AI));
```

$$AI := \begin{bmatrix} -\frac{2}{5} & 0 & -\frac{4}{5} \\ 0 & 0 & 0 \\ -\frac{4}{5} & 0 & -\frac{8}{5} \end{bmatrix}$$

$$1$$

$$2 \quad (2.4)$$

Symmetry compounded with translation. It has an invariant plane in the direction of the eigenspace associated to the eigenvalue 1, V(1). The normal vector to this plane is an eigenvector associated to the eigenvalue -1.

```
> X:=[x,y,z];
```

$$X := [x, y, z] \quad (2.5)$$

```
> Xf(X):=evalm(A&*X+b-X);
```

$$Xf([x, y, z]) := \begin{bmatrix} -\frac{2}{5}x - \frac{4}{5}z - \frac{1}{5} \\ 1 \\ -\frac{4}{5}x - \frac{8}{5}z + \frac{8}{5} \end{bmatrix} \quad (2.6)$$

```
> eigenvectors(A);
```

$$\left[1, 2, \left\{\left[\begin{array}{ccc} -2 & 0 & 1 \end{array}\right], \left[\begin{array}{ccc} 0 & 1 & 0 \end{array}\right]\right\}\right], \left[-1, 1, \left\{\left[\begin{array}{ccc} 1 & 0 & 2 \end{array}\right]\right\}\right] \quad (2.7)$$

The vector Xf(X) should be orthogonal to [1,0,2], condition that gives the equation of the invariant plane.

```
> pi:=innerprod([1,0,2],Xf(X))[1];
```

$$\pi := -2x - 4z + 3 \quad (2.8)$$

REMARKS:
1. This plane is parallel to the plane of the previous exercice, it has the same direction x1+2x3=0.
2. Both isometries have the same associated linear transformation.

We decompose the isometry given by M as a reflection with respect to the invariant plane compounded with a translation of vector parallel to the plane. The vector is obtained taking a point in the invariant plane and computing u=Qf(Q).

```
> solve(pi);
```

$$\left\{x = -2z + \frac{3}{2}, z = z\right\} \quad (2.9)$$

```
> Q:=[3/2,0,0];
```

$$Q := \left[\frac{3}{2}, 0, 0\right] \quad (2.10)$$

```
> QfQ:=evalm(A&*Q+b-Q);
```

$$QfQ := \begin{bmatrix} -\frac{4}{5} \\ 1 \\ \frac{2}{5} \end{bmatrix} \quad (2.11)$$

The matrix of the translation is:

```
> T:=matrix(4,4,[1,0,0,0,-4/5,1,0,0,1,0,1,0,2/5,0,0,1]);
```

$$T := \begin{bmatrix} 1 & 0 & 0 & 0 \\ -\frac{4}{5} & 1 & 0 & 0 \\ 1 & 0 & 1 & 0 \\ \frac{2}{5} & 0 & 0 & 1 \end{bmatrix} \quad (2.12)$$

M=T*S, then

```
> S:=evalm(inverse(T)&*M);
```

$$S := \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{3}{5} & \frac{3}{5} & 0 & -\frac{4}{5} \\ 0 & 0 & 1 & 0 \\ \frac{6}{5} & -\frac{4}{5} & 0 & -\frac{3}{5} \end{bmatrix} \quad (2.13)$$

```
> with(plots):
> gplane:=implicitplot3d(plano,x=-1..1,y=-1..1,z=-1..1,color=
  orange, style=surface):
> gu:=arrow(<-4/5,1,2/5>, shape = arrow,color=green):
> display(gplane,gu);
```




```
>
>
```

## Determine with respect to the orthonormal coordinate system R={O,B={$e_1$, $e_2$, $e_3$}} the equations of the reflection with respect to the line with equations $x+2z-2=0$ and $y=0$, that is a rotation with axis the line r and angle 180º.

```
> restart; with(linalg):
```

Matrix of the rotation of axis the line r in the coordinate system {Q, {u1,u2,u3}} with Q+<u1> the invariant line

```
> Gp:=matrix([[1,0,0,0],[0,1,0,0],[0,0,-1,0],[0,0,0,-1]]);
```

$$Gp := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix} \quad (3.1)$$

```
> solve({x+2*z-2 = 0,y = 0},{x,y,z});
```

$$\{x=-2z+2, y=0, z=z\} \quad (3.2)$$

```
> Q:=matrix(3,1,[2,0,0]); v1:=[-2,0,1];
```

$$Q := \begin{bmatrix} 2 \\ 0 \\ 0 \end{bmatrix}$$

$$v1 := [-2, 0, 1] \quad (3.3)$$

We look for a basis of the orthogonal space to v1

```
> solve({-2*x+z=0},{x,y,z});
```

$$\{x=x, y=y, z=2x\} \quad (3.4)$$

```
> v2:=[0,1,0]; v3:=[1,0,2];
```

$$v2 := [0, 1, 0]$$

$$v3 := [1, 0, 2] \quad (3.5)$$

```
> u[1]:= v1/(sqrt(innerprod(v1,v1)));u[2]:=v2/(sqrt(innerprod(v2,
  v2))); u[3]:=v3/(sqrt(innerprod(v3,v3)));
```

$$u_1 := \frac{1}{5}\,[-2, 0, 1]\sqrt{5}$$

$$u_2 := [0, 1, 0]$$

$$u_3 := \frac{1}{5}\,[1, 0, 2]\sqrt{5} \quad (3.6)$$

```
> U1:=matrix(3,1,evalm(u[1]));U2:=matrix(3,1,evalm(u[2]));U3:=
  matrix(3,1,evalm(u[3]));
```

$$U1 := \begin{bmatrix} -\frac{2}{5}\sqrt{5} \\ 0 \\ \frac{1}{5}\sqrt{5} \end{bmatrix}$$

$$U2 := \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}$$

$$U3 := \begin{bmatrix} \frac{1}{5}\sqrt{5} \\ 0 \\ \frac{2}{5}\sqrt{5} \end{bmatrix} \quad (3.7)$$

```
> RpR:=stackmatrix(matrix([[1,0,0,0]]),concat(Q,U1,U2,U3));
```

$$RpR := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 2 & -\frac{2}{5}\sqrt{5} & 0 & \frac{1}{5}\sqrt{5} \\ 0 & 0 & 1 & 0 \\ 0 & \frac{1}{5}\sqrt{5} & 0 & \frac{2}{5}\sqrt{5} \end{bmatrix} \quad (3.8)$$

```
> G:=evalm(RpR&*Gp&*inverse(RpR));
```

$$G := \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{4}{5} & \frac{3}{5} & 0 & -\frac{4}{5} \\ 0 & 0 & -1 & 0 \\ \frac{8}{5} & -\frac{4}{5} & 0 & -\frac{3}{5} \end{bmatrix} \quad (3.9)$$

```
> with(plots):
> line:=spacecurve([a,0,0],a=-2..2):
> rot:=spacecurve([1,cos(a),sin(a)],a=0..Pi,color=green):
> ge1:=arrow(<1,0,0>, shape = arrow,color=blue):
> ge2:=arrow(<0,1,0>, shape = arrow,color=blue):
> ge3:=arrow(<0,0,1>, shape = arrow,color=blue):
> display(line,rot,ge1,ge2,ge3);
```